Supporting Code for Aeronautical Journal Paper

This file consists of computational results and additional results about the matrices at the center of lifting line theory and the 2D vortex lattice model, supplementing the paper, “Revisiting the Combinatorics of Lifting Line and 2D Vortex Lattice Theory”

Initializations

Contains useful functions to simplify code execution below.

In[1]:=
```
$Assumptions = {nn ∈ Integers, nn ≥ 1, k ∈ Integers}
```

Out[1]=
```
{nn ∈ ℤ, nn ≥ 1, k ∈ ℤ}
```

slightly cleaner function for generating a matrix form of matrix

In[2]:=
```
mf[mat_] := MatrixForm[mat];
```

make a vector of 1’s

In[3]:=
```
makeEE[n_] := Table[{1}, n]
```

make the Q matrix

In[4]:=
```
makeQQ[n_] := Table[1/(1 - 4 (i - j)^2), {i, n}, {j, n}]
```

A little bit more terse version

In[5]:=
```
mq[n_] := makeQQ[n];
```

Make the inverse of the Q matrix. This will consist of rational numbers, which can get bogged down for large matrices. In that case, create Q, convert to numeric, and then invert.

In[6]:=
```
mqi[n_] := mq[n] // Inverse;
```

Sum the terms of a matrix

In[7]:=
```
su[mat_] := mat // Flatten // Total;
```

Get the row sums of a matrix

In[8]:=
```
rs[mat_] := mat // Transpose // Total;
```

Make the A matrix from James' paper

In[9]:=
$$\text{makeA[n\_] := Table}\left[\frac{1}{i - j + \frac{1}{2}}, \{i, n\}, \{j, n\}\right];$$

Make the H matrix from this paper

In[10]:=
$$\text{makeH[n\_] := Table}\left[\frac{1}{2\,(i - j) + 1}, \{i, n\}, \{j, n\}\right];$$

In[11]:=
```
SetDirectory["D:\\Shares\\Dropbox\\Research\\Sequential Model\\Aero Journal"]
```

... SetDirectory: Cannot set current directory to D:\Shares\Dropbox\Research\Sequential Model\Aero Journal.

Out[11]= $Failed

Some Q Properties

Q has elements $q_{i,j} = \frac{1}{1-4\,(i-j)^2}$

Q is an M-matrix, which implies many properties. It is also symmetric, which additionally constrains it. Wikipedia has a nice description.
https://en.wikipedia.org/wiki/M-matrix

Definition: Let A be a n × n real Z-matrix. That is, A = ($a_{i,j}$) where $a_{i,j} \leq 0$ for all i ≠ j, 1 ≤ i,j ≤ n. Then matrix A is also an M-matrix if it can be expressed in the form A = sI – B, where B = ($b_{i,j}$) with $b_{i,j} \geq 0$, for all 1 ≤ i,j ≤ n, where s is at least as large as the maximum of the moduli of the eigenvalues of B, and I is an identity matrix.

This results in a matrix with positive diagonal terms and the rest negative.

A couple of the more useful ones:
- Q is positive definite
- All eigenvalues are positive
- Its inverse is positive, $Q^{-1} \geq 0$, which states that all elements of Q^{-1} are greater than or equal to zero.
- There exists a positive diagonal matrix D such that AD has all positive row sums.

We can use the last property to show that Q is an M-matrix, by showing that Q always has positive row sums. The smallest row sum of Q will be when i=n/2, so

In[12]:=
```
midRowSum =
  (Assuming[n ∈ Integers, Sum[1/(1 - 4 (i - j)^2), {j, 1, n}]] /. i → n/2) // FullSimplify
```

Out[12]= $\frac{n}{-1 + n^2}$

```
Plot[midRowSum, {n, 1, 1000}, AxesLabel → {"n", "MidRowSum"}]
```




```
Limit[midRowSum, n → Infinity]
```

0

Look at Q as a variable sized matrix. Orange-red is positive, blue is negative. Left is Q, right is Q^{-1}

```
Manipulate[GraphicsRow[{Text["N = " <> ToString[n]],
   MatrixPlot[mq[n]], MatrixPlot[mq[n] // N // Inverse]}], {n, 10, 50, 5}]
```



Look at the row sums of Q converging to an ellipse

```
Manipulate[GraphicsRow[
  {Text["N = " <> ToString[n]], ListPlot[mq[n] // N // Inverse // rs]}], {n, 10, 50, 5}]
```



As an OBTW, the eigenstructure of Q is interesting. The eigenvalues can be shown to converge to the relationship $\lambda_i = \frac{\pi}{2}\,\text{Sin}\left[\frac{\pi}{2}\,\frac{i}{n+1}\right]$ where $\lambda_i$is the ith eigenvalue, *smallest to largest*.

In[∘]:=
```
count = 100;
```

In[∘]:=
```
eigsPlot = ListPlot[{Range[count]/(count + 1), Eigenvalues[mq[count] // N] // Reverse} // Transpose,
  Joined → True,
  Frame → True,
  FrameTicks -> {π/16 Range[8], π/8 Range[4]}, GridLines -> {π/16 Range[8], π/8 Range[4]}]
```

Out[∘]=



In[∘]:=
```
eigsApproxPlot = ListPlot[Table[{idx/(count + 1), π/2 Sin[π/2 idx/(count + 1)]}, {idx, 1, count}],
  Frame → True,
  FrameTicks -> {π/16 Range[8], π/8 Range[4]}, GridLines -> {π/16 Range[8], π/8 Range[4]}]
```

Out[∘]=



In[∘]:=
```
Show[eigsPlot, eigsApproxPlot]
```

Out[∘]=



The eigenvectors can be reasonably approximated by a discrete sine transform, with the kth eigenvector having ith terms $\sqrt{\frac{2}{n+1}}$ Sin$\left[\frac{i\,k\,\pi}{n+1}\right]$;

In[61]:=
```
count = 200;
idx = 5;
qq = mq[count] // N;
eveRaw = qq // Eigenvectors // Reverse;
eve = Map[Sign[# // First] × # &, eveRaw];
(* this is to ensure the eigenvectors have positive first components,
to match the DST terms *)
ss = Table[Sqrt[2/(count + 1)] Sin[(i k π)/(count + 1)], {i, 1, count}, {k, 1, count}]; // N;
(* discrete sine transform *)
Manipulate[ListPlot[{eve[[idx]], ss[[idx]]}, Joined → True, PlotLegends → {"Real", "Approx"},
  Frame → True, GridLines → {Automatic, Automatic}], {idx, 1, count, 1}]
```

Out[67]=



Bottom Line on Q

the ultimate result is...for the elements of the inverse...

In[13]:=
```
mqi[i_, j_, m_] := 1/(1 + 2 i - 2 j) 2^(2-4 m) j (1 - 2 j + 2 m) Binomial[2 i, i]
  Binomial[2 j, j] Binomial[2 (-j + m), -j + m] Binomial[-2 i + 2 m, -i + m]
  HypergeometricPFQ[{1, 1/2 + i, 1/2 + i - j, i - m}, {1 + i, 3/2 + i - j, 1/2 + i - m}, 1]
```

In[∘]:=
```
mqi[i, j, m] // TraditionalForm
```

Out[∘]//TraditionalForm=

$$\frac{j\,2^{2-4m}\binom{2i}{i}\binom{2j}{j}(-2j+2m+1)\binom{2m-2i}{m-i}\binom{2(m-j)}{m-j}\,{}_4F_3\!\left(1,\,i+\tfrac{1}{2},\,i-j+\tfrac{1}{2},\,i-m;\,i+1,\,i-j+\tfrac{3}{2},\,i-m+\tfrac{1}{2};\,1\right)}{2i-2j+1}$$

The ultimate expression for the row sums is

In[14]:=
```
mqirs[k_, m_] := (k (2 m - 2 k + 1))/2^(2 m-1) Binomial[2 k, k] Binomial[2 m - 2 k, m - k]
```

In[∘]:=
```
HoldForm[(k (2 m - 2 k + 1))/2^(2 m-1) Binomial[2 k, k] Binomial[2 m - 2 k, m - k]] // TraditionalForm
```

Out[∘]//TraditionalForm=

$$\frac{(k\,(2\,m-2\,k+1))\binom{2\,k}{k}\binom{2\,m-2\,k}{m-k}}{2^{2\,m-1}}$$

The sum of the elements of Q is

In[15]:=
```
qis[m_] := (m (m + 1))/2;
```

In[∘]:=
```
qis[m] // TraditionalForm
```

Out[∘]//TraditionalForm=

$$\frac{1}{2}\,m\,(m+1)$$

Demonstrate some of the equalities

In[∘]:=
```
count = 3;
```

In[∘]:=
```
qq = makeQQ[count];
qqi = mqi[count];
```

Show sum of elements of Q^{-1} = triangular number

In[∘]:=
```
Table[{i, mqi[i] // su}, {i, 1, 10}] // mf
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 1 \\ 2 & 3 \\ 3 & 6 \\ 4 & 10 \\ 5 & 15 \\ 6 & 21 \\ 7 & 28 \\ 8 & 36 \\ 9 & 45 \\ 10 & 55 \end{pmatrix}$$

show the row sums

In[∘]:=
```
qqi // rs
```

Out[∘]=

$$\left\{\frac{15}{8}, \frac{9}{4}, \frac{15}{8}\right\}$$

In[∘]:=
```
Table[mqirs[i, count], {i, 1, count}]
```

Out[∘]= $\left\{\frac{15}{8}, \frac{9}{4}, \frac{15}{8}\right\}$

show the elements

In[∘]:=
```
qqi // mf
```

Out[∘]//MatrixForm=
$$\begin{pmatrix} \frac{75}{64} & \frac{15}{32} & \frac{15}{64} \\ \frac{15}{32} & \frac{21}{16} & \frac{15}{32} \\ \frac{15}{64} & \frac{15}{32} & \frac{75}{64} \end{pmatrix}$$

In[∘]:=
```
Table[mqi[i, j, count], {i, 1, count}, {j, 1, count}] // mf
```

Out[∘]//MatrixForm=
$$\begin{pmatrix} \frac{75}{64} & \frac{15}{32} & \frac{15}{64} \\ \frac{15}{32} & \frac{21}{16} & \frac{15}{32} \\ \frac{15}{64} & \frac{15}{32} & \frac{75}{64} \end{pmatrix}$$

Sums of elements of Q and rows of Q (not inverted)

This result may or may not be useful. The row sum of the ith row is...

In[∘]:=
```
Assuming[{i, j, m} ∈ Integers,
  Sum[1/(1 - 4 (i - j)^2), {j, 1, m}] // FullSimplify] // TraditionalForm
```

Out[∘]//TraditionalForm=
$$\frac{m}{(2\,i-1)\,(-2\,i+2\,m+1)}$$

Getting the sum of all of the elements requires a bit more work, and some trickery to get Mathematica to give us an answer. Here is the brute force attempt. Mathematica can’t quite get it.

In[∘]:=
```
Assuming[{i, j, m} ∈ Integers,
  Sum[1/(1 - 4 (i - j)^2), {j, 1, m}, {i, 1, m}] // FullSimplify] // TraditionalForm
```

Out[∘]//TraditionalForm=
$$\sum_{j=1}^{m}\sum_{i=1}^{m}\frac{1}{1-4\,(i-j)^2}$$

Manipulate it a bit so the equation does the summation along Q's diagonals in order to gain an answer

In[∘]:=
```
res = Assuming[{{m} ∈ Integers}, m + Sum[2 (m - k + 1) 1/(1 - 4 (1 - k)^2), {k, 2, m}] // FullSimplify]
```

Out[∘]= $\frac{1}{2}\left(\text{EulerGamma} + \frac{2}{-1+2\,\text{m}} + \text{Log}[4] + \text{PolyGamma}\left[0, -\frac{1}{2}+\text{m}\right]\right)$

In[∘]:= `res // TraditionalForm`

Out[∘]//TraditionalForm=

$$\frac{1}{2}\left(\frac{2}{2\,m-1}+\psi^{(0)}\!\left(m-\frac{1}{2}\right)+\gamma+\log(4)\right)$$

Try it out.

In[∘]:= `res /. m → 55`

Out[∘]= $\frac{1}{2}\left(\frac{2}{109}+\text{EulerGamma}+\text{Log}[4]+\text{PolyGamma}\left[0,\frac{109}{2}\right]\right)$

In[∘]:= `res /. m → 55 // N`

Out[∘]= 2.98543

In[∘]:= `mq[55] // su // N`

Out[∘]= 2.98543

Bottom Line on A and H, form the 2D Vortex Lattice Model

Elements of the inverse of H matrix, where q = $\frac{1}{2}\left(H+H^T\right) = H\otimes H^T$ (Hadamard product)

In[16]:= $\text{mhi}[i\_, j\_, m\_] := \frac{1}{(-1+2\,i)\,(1-2\,i+2\,j)} 2^{2-4\,m}\, i\, j\, (1-2\,i+2\,m)\, \text{Binomial}[2\,i, i]\, \text{Binomial}[2\,j, j]\, \text{Binomial}[-2\,i+2\,m, -i+m]\, \text{Binomial}[-2\,j+2\,m, -j+m]$

In[∘]:= `mhi[i, j, m] // TraditionalForm`

Out[∘]//TraditionalForm=

$$\frac{i\,j\,2^{2-4\,m}\binom{2\,i}{i}\binom{2\,j}{j}(-2\,i+2\,m+1)\binom{2\,m-2\,i}{m-i}\binom{2\,m-2\,j}{m-j}}{(2\,i-1)\,(-2\,i+2\,j+1)}$$

The rows sum of the inverse of H

In[17]:= $\text{mhirs}[k\_, m\_] := \frac{2^{1-2\,m}\, k\, (1-2\,k+2\,m)\, \text{Binomial}[2\,k, k]\, \text{Binomial}[-2\,k+2\,m, -k+m]}{-1+2\,k}$

In[∘]:= $\frac{2^{1-2\,m}\, k\, (1-2\,k+2\,m)\, \text{Binomial}[2\,k, k]\, \text{Binomial}[-2\,k+2\,m, -k+m]}{-1+2\,k}$ `// TraditionalForm`

Out[∘]//TraditionalForm=

$$\frac{k\,2^{1-2\,m}\binom{2\,k}{k}(-2\,k+2\,m+1)\binom{2\,m-2\,k}{m-k}}{2\,k-1}$$

The rows sum of the inverse of H

In[18]:= $\text{mairs}[m\_, k\_] := \frac{2^{-2\,m}\, k\, (1-2\,k+2\,m)\, \text{Binomial}[2\,k, k]\, \text{Binomial}[-2\,k+2\,m, -k+m]}{-1+2\,k}$

In[∘]:=
```
mairs[m, k] // TraditionalForm
```
Out[∘]//TraditionalForm=

$$\frac{k\, 2^{-2 m} \binom{2 k}{k} (-2 k + 2 m + 1) \binom{2 m - 2 k}{m - k}}{2 k - 1}$$

Sum of elements of H^{-1}

In[19]:=
```
his[m_] := m
```

In[∘]:=
```
his[m] // TraditionalForm
```
Out[∘]//TraditionalForm=

$$m$$

Inverse of A matrix, where q = $\left(A + A^T\right) = \frac{1}{2} A \otimes A^T$ (Hadamard product)

In[20]:=
```
mai[i_, j_, m_] := 1/((-1 + 2 i) (1 - 2 i + 2 j)) 4^(1-2 m) i j (1 - 2 i + 2 m) Binomial[2 i, i]
   Binomial[2 j, j] Binomial[-2 i + 2 m, -i + m] Binomial[-2 j + 2 m, -j + m]
```

In[∘]:=
```
mai[i, j, m] // TraditionalForm
```
Out[∘]//TraditionalForm=

$$\frac{i\, j\, 4^{1-2 m} \binom{2 i}{i} \binom{2 j}{j} (-2 i + 2 m + 1) \binom{2 m - 2 i}{m - i} \binom{2 m - 2 j}{m - j}}{(2 i - 1) (-2 i + 2 j + 1)}$$

The rows sum of the inverse

In[21]:=
```
mhirs[k_, m_] := (2^(1-2 m) k (1 - 2 k + 2 m) Binomial[2 k, k] Binomial[-2 k + 2 m, -k + m])/(-1 + 2 k)
```

In[∘]:=
```
mhirs[k, m] // TraditionalForm
```
Out[∘]//TraditionalForm=

$$\frac{k\, 2^{1-2 m} \binom{2 k}{k} (-2 k + 2 m + 1) \binom{2 m - 2 k}{m - k}}{2 k - 1}$$

Sum of elements of H

In[22]:=
```
his[m_] := m
```

In[∘]:=
```
his[m] // TraditionalForm
```
Out[∘]//TraditionalForm=

$$m$$

Demonstrate some of the equalities

In[∘]:=
```
count = 3;
```

In[∘]:=
```
hh = makeH[count];
qq = makeQQ[count];
```

demonstrate on small matrix, show Q = $\frac{1}{2}\left(H + H^T\right)$

In[∘]:=
```
1/2 (hh + (hh // Transpose)) - qq // mf
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Hadamard Product Q = $\frac{1}{2}\left(H \otimes H^T\right)$

In[∘]:=
```
(hh × (hh // Transpose)) - qq // mf
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

verify inverse of H, first actually invert it.

In[∘]:=
```
(hhi = hh // Inverse) // mf
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \frac{45}{64} & \frac{15}{32} & \frac{45}{64} \\ -\frac{9}{32} & \frac{9}{16} & \frac{15}{32} \\ -\frac{3}{64} & -\frac{9}{32} & \frac{45}{64} \end{pmatrix}$$

then run the function for the elements of the inverse

In[∘]:=
```
Table[mhi[i, j, count], {i, count}, {j, count}] // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \frac{45}{64} & \frac{15}{32} & \frac{45}{64} \\ -\frac{9}{32} & \frac{9}{16} & \frac{15}{32} \\ -\frac{3}{64} & -\frac{9}{32} & \frac{45}{64} \end{pmatrix}$$

verify row sums of H

In[∘]:=
```
hhi // rs
```

Out[∘]= $\left\{\frac{15}{8}, \frac{3}{4}, \frac{3}{8}\right\}$

In[∘]:=
```
Table[mhirs[i, 3], {i, 3}]
```

Out[∘]= $\left\{\frac{15}{8}, \frac{3}{4}, \frac{3}{8}\right\}$

sums of elements of A and rows of A (not inverted)

In[∘]:=
```
rowsumsA =
  Assuming[{i, m} ∈ Integers, Sum[1/((i - j) + 1/2), {i, 1, m}] // FullSimplify] // TraditionalForm
```

Out[∘]//TraditionalForm=

$$H_{-j+m+\frac{1}{2}} - H_{\frac{1}{2}-j}$$

Mathematica bludgeons itself trying to get the full summation

In[∘]:=
```
Assuming[m ∈ Integers, Sum[rowsumsA, {j, 1, m}]] // FullSimplify
```

Out[∘]=

$$\sum_{j=1}^{m} \left(H_{-j+m+\frac{1}{2}} - H_{\frac{1}{2}-j}\right)$$

but we can eyeball the matrix and see a path to a solution. The sum of all the elements is just the sum of the last row (or first column)

```
makeA[5] // mf
```

$$\begin{pmatrix} 2 & -2 & -\frac{2}{3} & -\frac{2}{5} & -\frac{2}{7} \\ \frac{2}{3} & 2 & -2 & -\frac{2}{3} & -\frac{2}{5} \\ \frac{2}{5} & \frac{2}{3} & 2 & -2 & -\frac{2}{3} \\ \frac{2}{7} & \frac{2}{5} & \frac{2}{3} & 2 & -2 \\ \frac{2}{9} & \frac{2}{7} & \frac{2}{5} & \frac{2}{3} & 2 \end{pmatrix}$$

```
makeA[5] // su
```

$$\frac{1126}{315}$$

```
makeA[5] // Last // su
```

$$\frac{1126}{315}$$

```
1126/315 // N
```

3.5746

In[∘]:=
```
sumelementsA = Assuming[{i, m} ∈ Integers, Sum[1/((i - 1) + 1/2), {i, 1, m}] // FullSimplify]
```

Out[∘]= HarmonicNumber$\left[-\frac{1}{2} + m\right]$ + Log[4]

In[∘]:=
```
sumelementsA /. m → 5 // Simplify
```

Out[∘]= HarmonicNumber$\left[\frac{9}{2}\right]$ + Log[4]

In[∘]:=
```
sumelementsA // TraditionalForm
```

Out[∘]//TraditionalForm=

$$H_{m-\frac{1}{2}} + \log(4)$$

Q and H interactions

Inverse of H matrix, where q = $\frac{1}{2}\left(H + H^T\right) = H \otimes H^T$ (Hadamard product), not too much simpler

The rows sum of the inverse really squishes down

In[∘]:=
```
Assuming[k > 1, mqirs[k, m] / mairs[k, m] // FullSimplify] // TraditionalForm
```

Out[∘]//TraditionalForm=

$$\frac{4\,\Gamma\left(k+\frac{1}{2}\right)\Gamma(m)\,\Gamma(k-m+1)\,\Gamma\left(-k+m+\frac{3}{2}\right)}{\Gamma(k)\,\Gamma\left(m-\frac{1}{2}\right)\Gamma\left(k-m+\frac{3}{2}\right)\Gamma(-k+m+1)}$$

Sum of elements of H

look at approximations of row sums

In[∘]:=
```
qirsa[n_, k_] := 4/π Sqrt[k (n + 1/2 - k)]
```

$$\text{qirsa[n\_, k\_]} := \frac{4}{\pi}\sqrt{k\left(n+\frac{1}{2}-k\right)}$$

In[∘]:=
```
hirsa[n_, k_] := (2 Sqrt[2] Sqrt[k (2 n - 2 k + 1)]) / (π (2 k - 1))
```

$$\text{hirsa[n\_, k\_]} := \frac{2\sqrt{2}\sqrt{k\,(2n-2k+1)}}{\pi\,(2k-1)}$$

In[∘]:=
```
qirsa[m, k] / hirsa[m, k] // Simplify // TraditionalForm
```

Out[∘]//TraditionalForm=

$$2k-1$$

Approximation to the Central Binomial Terms

This is notes screen captured from the Wikipedia page on central binomial coefficients. Lots of interesting information there.
https://en.wikipedia.org/wiki/Central_binomial_coefficient

They also satisfy the recurrence

$$\binom{2(n+1)}{n+1} = \frac{4n+2}{n+1}\cdot\binom{2n}{n}.$$

The Wallis product can be written in asymptotic form for the central binomial coefficient:

$$\binom{2n}{n} \sim \frac{4^n}{\sqrt{\pi n}}.$$

By Wallis's formula, both middle expressions converge to $\frac{2}{\pi}$. The right hand side of the first equation is increasing, while the right hand side of the second equation is decreasing. We conclude that

$$\frac{1}{\sqrt{\pi(n+1/2)}} \leq \binom{2n}{n}\frac{1}{4^n} \leq \frac{1}{\sqrt{\pi n}}.$$

so for terms of $\binom{2k}{k}\binom{2n-2k}{n-k}$ we can approximate a couple of different ways. Use the upper bound for the left term and lower bound for the right to get
$\binom{2k}{k}\binom{2n-2k}{n-k} \simeq \frac{2^{2n}}{\pi\sqrt{k(n-k+1/2)}} = \frac{2^{2n}\sqrt{2}}{\pi\sqrt{k(2n-2k+1)}}$

Plot of Q^{-1} row sums and approximation

In[∘]:=
```
qRowInverseApprox[k_, nn_] := 4/π Sqrt[(k - 1/2) (nn - k + 1/2)]
```

In[∘]:=
```
count = 200;
```

In[∘]:=
```
qRowInverse = {Range[count], mq[count] // N // Inverse // rs} // Transpose;
```

In[∘]:=
```
qRowSumsPlot = Plot[qRowInverseApprox[k, count], {k, 1, count},
  PlotRange → All,
  Epilog → {PointSize[Medium], Point[qRowInverse[[1 ;; count ;; 5]]]},
  Frame → True,
  GridLines → Automatic,
  PlotStyle → {Black},
  Frame → True,
  GridLines → Automatic,
  FrameLabel → {Style["Index", Black], Style["Row sum", Black]},
  PlotLabel → Style["Row sums of Q^-1", FontColor → Black]
 ]
```

Out[∘]=



THE APPROXIMATION to Row Sums of H and A

In[]:=
$$\text{hrowAPPROX[nn\_, k\_]} := \frac{4}{\pi}\,\frac{\sqrt{k\left(nn - k + \frac{1}{2}\right)}}{(2k-1)}$$

In[]:=
$$\text{arowAPPROX[nn\_, k\_]} := \frac{2}{\pi}\,\frac{\sqrt{k\left(nn - k + \frac{1}{2}\right)}}{(2k-1)}$$

In[]:=
```
count = 200;
```

In[]:=
```
hRowSums = {Range[count], makeH[count] // N // Inverse // Transpose // Total} // Transpose;
```

In[]:=
```
hrowsumsplot = Plot[hrowAPPROX[count, k], {k, 1, count},
  Epilog → {PointSize[Medium], Point[hRowSums[[1 ;; count ;; 5]]]},
  Frame → True,
  GridLines → Automatic,
  PlotStyle → {Black},
  Frame → True,
  GridLines → Automatic,
  FrameLabel → {Style["Index", Black], Style["Row sum", Black]},
  PlotLabel → Style["Row sums of H^-1", FontColor → Black]]
```

Out[]=



Wing Modeling - Simple Example

All of the calculations below break the wing into *n* segments, where *n* is set here, and the common terms (a, C, I, e, Q, W, Q^{-1}, W^{-1}) are calculated. Use sparse arrays for all diagonal matrices.

In[35]:=
```
n = 100;
ii = SparseArray@IdentityMatrix[n];
a = 2 π;
```

Generate the matrices. Note that Q is made up of rational terms, but is converted to numerical in order to easily invert it.

Also, the e vector is a true vector in the sense that it is a $n \times 1$ matrix. The only place this potentially offers confusion is when we compute $e^T Q e$ or similar. This results in a 1×1 matrix as opposed to a scalar.

In[38]:=
```
qq = mq[n] // N;
ee = makeEE[n];
qqi = Inverse@(qq // N);
ww = (a n)/(2 π) qq // N;
wwi = Inverse@ww;
```

This helper function takes a shape for the chords, and scales it to match an aspect ratio AR

In[43]:=
```
scaleC[cc_, AR_] := Length[cc]/AR cc/su[cc];
```

This function computes the Gamma vector, taking in a vector of chords cc and vector α. We assume ρ and Velocity = 1

Although the form is $\Gamma = e^T(C^{-1} + W)^{-1}\alpha$ we do not find the inverse explicitly. Instead we use Linear-Solve[ ] to find x such that $(C^{-1} + W)x = \alpha$ and then compute $e^T x$

```
gammaCalc[cc_, α_] := a 1/2 LinearSolve[(DiagonalMatrix@SparseArray@1/cc + ww), α];
```

Blocky Wing AR = 10

In this setup, the wing has a jump in chord length towards the wingtips. Note the use of sparse arrays for the diagonal matrices

In[46]:=
```
AR = 10;
```

Make the wing

In[47]:=
```
ccint = Table[{ 2   Abs[k - n/2] > n/4
              { 1   True          , {k, 1, n}];
```

In[48]:=
```
ccBlock = scaleC[ccint, AR];
```

In[49]:=
```
ListPlot[Transpose[{Range[n]/n, ccBlock}],
 Joined → True,
 AspectRatio → Automatic,
 Axes → False,
 Filling → Axis,
 FillingStyle → Black,
 PlotStyle → White]
```

Out[49]=

In[50]:=
```
αVec = Table[1, {n}];
```

Show the lift distribution (which is also the Γ distribution)

In[60]:=
```
ListPlot[{gammaCalc[ccBlock, αVec / 30] + .005}, Joined → True,
 Frame → True,
 PlotStyle → Black,
 FrameLabel → {Style["Index", Black], Style["Γ", Black]}]
```

Out[60]=



Now vary the aspect ratio and see how c_L changes, using $c_L = \frac{\mathbf{e}^T \left(C^{-1} + W\right)^{-1}}{\mathbf{e}^T C \mathbf{e}} a \boldsymbol{\alpha}$

In[52]:=
```
clValues = Table[{AR, 2 a su@gammaCalc[scaleC[ccint, AR], αVec / 30] / su@scaleC[ccint, AR]}, {AR, 7, 70}];
```

In[53]:=
```
ListPlot[clValues,
 Joined → True,
 Frame → True,
 GridLines → Automatic,
 PlotStyle → Black,
 FrameLabel → {Style["AR", Black], Style["cL", Black]}]
```

Out[53]=



PROOF OF PROPERTIES OF Q

This section requires the Fast Zeilberger package to run. It can be acquired from https://www3.risc.jku.at/research/combinat/software/ergosum/RISC/fastZeil.html

The site includes references and instructions. If you load it correctly, you should see the following:

In[24]:=
```
<< RISC`fastZeil`
```

Fast Zeilberger Package version 3.61
written by Peter Paule, Markus Schorn, and Axel Riese
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

```
<< RISC`fastZeil`
```

Fast Zeilberger Package version 3.61
written by Peter Paule, Markus Schorn, and Axel Riese
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

Proof that the elements of Q^{-1} sum to the triangular number *T(n)*

This proof actually proves that the sum of the row sums of Q^{-1} is hte triangular number T(n), and so depends on the proof of the row sums term. First define the row sums expression (taken from above), replace the formal binomial[ ] function with the factorials)

rowSumsExpression = $\frac{k\,(2\,nn-2\,k+1)}{2^{2\,nn-1}}\left(\frac{(2\,k)\,!}{k\,!\;k\,!}\right)\left(\frac{(2\,nn-2\,k)\,!}{(nn-k)\,!\;(nn-k)\,!}\right)$

$$\frac{2^{1-2\,nn}\,k\,(1-2\,k+2\,nn)\,(2\,k)\,!\,(-2\,k+2\,nn)\,!}{(k\,!)^2\,((-k+nn)\,!)^2}$$

We are trying to show that the sum of the row sums satisfies

HoldForm[**Sum**[$\frac{k\,(2\,nn-2\,k+1)}{2^{2\,nn-1}}\left(\frac{(2\,k)\,!}{k\,!\;k\,!}\right)\left(\frac{(2\,nn-2\,k)\,!}{(nn-k)\,!\;(nn-k)\,!}\right)$, {k, 1, nn}] == $\frac{nn\,(nn+1)}{2}$] //
TraditionalForm

$$\sum_{k=1}^{nn}\frac{(k\,(2\,nn-2\,k+1))\,(2\,k)!\,(2\,nn-2\,k)!}{2^{2\,nn-1}\,(k!\,k!)\,((nn-k)!\,(nn-k)!)}=\frac{1}{2}\,nn\,(nn+1)$$

Or moving the right hand side terms to the left, and into the summation

HoldForm[**Sum**[$\frac{1}{\frac{nn\,(nn+1)}{2}}\;\frac{k\,(2\,nn-2\,k+1)}{2^{2\,nn-1}}\left(\frac{(2\,k)\,!}{k\,!\;k\,!}\right)\left(\frac{(2\,nn-2\,k)\,!}{(nn-k)\,!\;(nn-k)\,!}\right)$, {k, 1, nn}] == 1] //
TraditionalForm

$$\sum_{k=1}^{nn}\frac{(k\,(2\,nn-2\,k+1))\,(2\,k)!\,(2\,nn-2\,k)!}{\frac{1}{2}\,(nn\,(nn+1))\,2^{2\,nn-1}\,(k!\,k!)\,((nn-k)!\,(nn-k)!)}=1$$

define the summand as f(n,k)

fnk = $\frac{2}{nn\,(nn+1)}$ **rowSumsExpression**

$$\frac{2^{2-2\,nn}\,k\,(1-2\,k+2\,nn)\,(2\,k)\,!\,(-2\,k+2\,nn)\,!}{nn\,(1+nn)\,(k\,!)^2\,((-k+nn)\,!)^2}$$

Call the zeilberger algorithm to exercise the proof. Immediately returned in the recurrence

```
res = Zb[fnk, {k, 1, nn}, nn, 1]
If `-1 + nn' is a natural number, then:
{SUM[nn] - SUM[1 + nn] == 0}
```

show that the base case is correct

```
fnk /. {nn → 1, k → 1}
1
```

Get the “proof certificate”. This is a value set once you’ve run the Zb[ ] function

```
rnk = show[R]
```

$$\frac{(-1+k)\ (3-2\,k+2\,nn)}{2\ (2+nn)\ (1-k+nn)}$$

Here we can ask for a full descriptor of the proof, which is available once you've run the Zb[ ] function. It generates a new notebook, contents of which are copied below.

```
Prove[]
```

Theorem

Let

$$F(k,\ \text{nn}) = \frac{k\,2^{2-2\,\text{nn}}\,(2\,k)!\,(-2\,k+2\,\text{nn}+1)\,(2\,\text{nn}-2\,k)!}{\text{nn}\,(\text{nn}+1)\,(k!)^2\,((\text{nn}-k)!)^2}$$

and

$$\text{SUM}(\text{nn}) = \sum_{k=1}^{\text{nn}} F(k,\ \text{nn}).$$

If `-1 + nn' is a natural number, then:

$$\text{SUM}(\text{nn}) - \text{SUM}(1+\text{nn}) = 0.$$

Proof

Let Δ_k(".") denote the forward difference operator in *k* and define

$$R(k,\ \text{nn}) = \frac{(k-1)\,(-2\,k+2\,\text{nn}+3)}{2\,(\text{nn}+2)\,(-k+\text{nn}+1)}.$$

Then the Theorem follows from summing the equation

$$F(k,\ \text{nn}) - F(k,\ 1+\text{nn}) = \Delta_k(F(k,\ \text{nn})\,R(k,\ \text{nn}))$$

over *k* from 1 to nn + 1.

This equation is routinely verifiable by dividing both sides by *F* and checking the resulting rational equation.

Verify

```
show[R]
```

$$\frac{(-1+k)\ (3-2\,k+2\,nn)}{2\ (2+nn)\ (1-k+nn)}$$

this is plain text, so copy into an input cell

```
(rnk = ((-1 + k) (3 - 2 k + 2 nn)) / (2 (2 + nn) (1 - k + nn)))
```

$$\frac{(-1+k)\,(3-2\,k+2\,nn)}{2\,(2+nn)\,(1-k+nn)}$$

We need g(n,k) and g(n,k+1). Go ahead and divide out the f(n,k) now to avoid the factorials

```
gnk = (fnk rnk) / fnk
```

$$\frac{(-1+k)\,(3-2\,k+2\,nn)}{2\,(2+nn)\,(1-k+nn)}$$

```
gnkplus1 = ((fnk rnk) /. k → k + 1) / fnk // FullSimplify
```

$$\frac{1+2\,k}{4+2\,nn}$$

Then we need f(n,k) and f(n+1,k)

```
fnknplus1 = (fnk /. nn → nn + 1) / fnk // FullSimplify
```

$$\frac{(-3+2\,k-2\,nn)\,nn}{2\,(-1+k-nn)\,(2+nn)}$$

Show that $\frac{f(n,k)-f(n+1,k)}{f(n,k)} == \frac{g(n,k+1)=g(n,k)}{f(n,k)}$

```
1 - fnknplus1 == gnkplus1 - gnk // FullSimplify
```

```
True
```

Show that the g(n,k) goes to zero on both ends

```
(fnk rnk) /. k → 0
```

```
0
```

```
(fnk rnk) /. k → nn + 2
```

```
0
```

Proof that Row sums are correct

The idea behind the proof is that (letting e be a vector of 1's)

$$Q\,Q^{-1} = I$$
$$Q\,Q^{-1}\,e = e$$
$$Q\,(Q^{-1}\,e) = e$$

The term $Q^{-1} e$ is the vector of row sums, call it r. To show that it is correct, for each row (or column, since symmetric) q in Q, we need $q^T r = 1$

this is the expression for the sum of the kth row of Q^{-1}.

rowSumsExpression = $\frac{k\,(2\,nn-2\,k+1)}{2^{2\,nn-1}}\left(\frac{(2\,k)\,!}{k\,!\;k\,!}\right)\left(\frac{(2\,nn-2\,k)\,!}{(nn-k)\,!\;(nn-k)\,!}\right)$**;**

This is the summand that corresponds to $q^T r = 1$ for the ith row. We will prove it for all i

fnk = $\frac{1}{1-4\,(i-k)^2}$ **rowSumsExpression**

$$\frac{2^{1-2\,nn}\,k\,(1-2\,k+2\,nn)\,(2\,k)\,!\,(-2\,k+2\,nn)\,!}{(1-4\,(i-k)^2)\,(k\,!)^2\,((-k+nn)\,!)^2}$$

Run the algorithm. Note the last argument is a ‘2’ instead of a ‘1’ like up above. This is necessary, as the algorithm has to generate a recurrence in two terms instead of one. Something you have to play with.

```
res = Zb[fnk, {k, 1, nn}, nn, 2] // First
```

```
If `-1 + nn' is a natural number, then:
```

$$2\,(-1+i-nn)\,(2+nn)\,\text{SUM}[nn]+(9-6\,i+13\,nn-4\,i\,nn+4\,nn^2)\,\text{SUM}[1+nn]+(-5+2\,i-2\,nn)\,(1+nn)\,\text{SUM}[2+nn]==0$$

this requires a little extra effort to get the induction going. Need base cases of *nn=1 and nn=2*

First base case

```
base1 = fnk /. {nn → 1, i → 1, k → 1}
```

1

Now base case for *n*=2, we need to consider *i*=1,2

```
base2 = Sum[fnk /. nn → 2, {k, 1, 2}]
```

$$\frac{3}{2\,(1-4\,(-2+i)^2)}+\frac{3}{2\,(1-4\,(-1+i)^2)}$$

```
base2 /. i → 1
```

1

```
base2 /. i → 2
```

1

Show that when SUM[nn] and SUM[nn+1] equal one, then SUM[nn+2] -> 1, for all *i*

```
Solve[2 (-1 + i - nn) (nn + 2) sumN +
    (9 - 6 i + 13 nn - 4 i nn + 4 nn^2) sumNplus1 + (-5 + 2 i - 2 nn) (1 + nn) sumNplus2 == 0,
  sumNplus2] /. {sumN → 1, sumNplus1 → 1} // FullSimplify
```

```
{{sumNplus2 → 1}}
```

now move on to the proof, after showing g(n,k) is bounded by zero. Get proof certificate

```
rnk = show[R]
```

$$\frac{(-1+k)\,(1-2\,i+2\,k)\,(3-2\,k+2\,nn)}{4\,(1-k+nn)\,(2-k+nn)}$$

g (n, 1) = 0

```
(rnk fnk) /. k → 1
```

```
0
```

g (n, n+3) = 0

```
(rnk fnk) /. k → nn + 3
```

```
0
```

Generate the proof, which creates a new notebook, contents copied below:

```
Prove[]
```

Theorem

Let

$$F(k,\ \mathrm{nn}) = \frac{k\,2^{1-2\,\mathrm{nn}}\,(2\,k)!\,(2\,k-2\,\mathrm{nn}-1)\,(2\,\mathrm{nn}-2\,k)!}{(k!)^2\,(2\,i-2\,k-1)\,(2\,i-2\,k+1)\,((\mathrm{nn}-k)!)^2}$$

and

$$\mathrm{SUM}(\mathrm{nn}) = \sum_{k=1}^{\mathrm{nn}} F(k,\ \mathrm{nn}).$$

If \`-1 + nn' is a natural number, then:

$$\left(-4\,i\,\mathrm{nn} - 6\,i + 4\,\mathrm{nn}^2 + 13\,\mathrm{nn} + 9\right)\mathrm{SUM}(1+\mathrm{nn}) + 2\,(\mathrm{nn}+2)\,(i-\mathrm{nn}-1)\,\mathrm{SUM}(\mathrm{nn}) + (\mathrm{nn}+1)\,(2\,i-2\,\mathrm{nn}-5)\,\mathrm{SUM}(2+\mathrm{nn}) = 0.$$

Proof

Let $\Delta_k(".")$ denote the forward difference operator in k and define

$$R(k,\ \mathrm{nn}) = \frac{(k-1)\,(-2\,i+2\,k+1)\,(-2\,k+2\,\mathrm{nn}+3)}{4\,(-k+\mathrm{nn}+1)\,(-k+\mathrm{nn}+2)}.$$

Then the Theorem follows from summing the equation

$$\left(-4\, i\, \mathrm{nn} - 6\, i + 4\, \mathrm{nn}^2 + 13\, \mathrm{nn} + 9\right) F(k,\ 1 + \mathrm{nn}) + 2\,(\mathrm{nn} + 2)\,(i - \mathrm{nn} - 1)\, F(k,\ \mathrm{nn}) + (\mathrm{nn} + 1)\,(2\, i - 2\, \mathrm{nn} - 5)\, F(k,\ 2 + \mathrm{nn}) = \Delta_k(F(k,\ \mathrm{nn})\, R(k,\ \mathrm{nn}))$$

over k from 1 to nn + 2.

This equation is routinely verifyable by dividing both sides by F and checking the resulting rational equation.

Verify

```
gnk = rnk fnk
```

$$\frac{2^{-1-2\,\mathrm{nn}}\,(-1+k)\,k\,(1-2\,i+2\,k)\,(1-2\,k+2\,\mathrm{nn})\,(3-2\,k+2\,\mathrm{nn})\,(2\,k)!\,(-2\,k+2\,\mathrm{nn})!}{\left(1-4\,(i-k)^2\right)\,(1-k+\mathrm{nn})\,(2-k+\mathrm{nn})\,(k!)^2\,((-k+\mathrm{nn})!)^2}$$

```
gnkplus1 = gnk /. k → k + 1
```

$$\left(2^{-1-2\,\mathrm{nn}}\,k\,(1+k)\,(1-2\,i+2\,(1+k))\,(1-2\,(1+k)+2\,\mathrm{nn})\,(3-2\,(1+k)+2\,\mathrm{nn})\,(2\,(1+k))!\,(-2\,(1+k)+2\,\mathrm{nn})!\right) \Big/ \left(\left(1-4\,(-1+i-k)^2\right)\,(-k+\mathrm{nn})\,(1-k+\mathrm{nn})\,((1+k)!)^2\,((-1-k+\mathrm{nn})!)^2\right)$$

```
fnknplus1 = fnk /. nn → nn + 1
```

$$\frac{2^{1-2\,(1+\mathrm{nn})}\,k\,(1-2\,k+2\,(1+\mathrm{nn}))\,(2\,k)!\,(-2\,k+2\,(1+\mathrm{nn}))!}{\left(1-4\,(i-k)^2\right)\,(k!)^2\,((1-k+\mathrm{nn})!)^2}$$

```
fnknplus2 = fnk /. nn → nn + 2
```

$$\frac{2^{1-2\,(2+\mathrm{nn})}\,k\,(1-2\,k+2\,(2+\mathrm{nn}))\,(2\,k)!\,(-2\,k+2\,(2+\mathrm{nn}))!}{\left(1-4\,(i-k)^2\right)\,(k!)^2\,((2-k+\mathrm{nn})!)^2}$$

relationship is

$$(-4\, i\, \mathrm{nn} - 6\, i + 4\, \mathrm{nn}^2 + 13\, \mathrm{nn} + 9)\, F(k,\ 1 + \mathrm{nn}) + 2\,(\mathrm{nn} + 2)\,(i - \mathrm{nn} - 1)\, F(k,\ \mathrm{nn}) + (\mathrm{nn} + 1)\,(2\, i - 2\, \mathrm{nn} - 5)\, F(k,\ 2 + \mathrm{nn}) = \Delta_k(F(k,\ \mathrm{nn})\, R(k,\ \mathrm{nn}))$$

$$\left(-4\, i\, \mathrm{nn} - 6\, i + 4\, \mathrm{nn}^2 + 13\, \mathrm{nn} + 9\right) F(k,\ 1 + \mathrm{nn}) + 2\,(\mathrm{nn} + 2)\,(i - \mathrm{nn} - 1)\, F(k,\ \mathrm{nn}) + (\mathrm{nn} + 1)\,(2\, i - 2\, \mathrm{nn} - 5)\, F(k,\ 2 + \mathrm{nn}) = \Delta_k(F(k,\ \mathrm{nn})\, R(k,\ \mathrm{nn}))$$

$$2\,(\mathrm{nn}+2)\,(i-\mathrm{nn}-1)\,\frac{\mathrm{fnk}}{\mathrm{fnk}} + \left(-4\,i\,\mathrm{nn}-6\,i+4\,\mathrm{nn}^2+13\,\mathrm{nn}+9\right)\frac{\mathrm{fnknplus1}}{\mathrm{fnk}} + (\mathrm{nn}+1)\,(2\,i-2\,\mathrm{nn}-5)\,\frac{\mathrm{fnknplus2}}{\mathrm{fnk}} == \frac{\mathrm{gnkplus1}}{\mathrm{fnk}} - \frac{\mathrm{gnk}}{\mathrm{fnk}} \ \text{// FullSimplify}$$

```
True
```

proof of Properties of H

Follows the same order as the proof for properties of Q

prove that the sum of the row sums add to N, or $e^T H^{-1} e = n$

here is the term we are proving

```
HoldForm[Sum[1/nn (2^(1-2 nn) k (1 - 2 k + 2 nn) Binomial[2 k, k] Binomial[-2 k + 2 nn, -k + nn])/(-1 + 2 k),
   {k, 1, nn}]] // TraditionalForm
```

$$\sum_{k=1}^{nn} \frac{2^{1-2\,nn}\,k\,(1-2\,k+2\,nn)\binom{2\,k}{k}\binom{-2\,k+2\,nn}{-k+nn}}{nn\,(-1+2\,k)}$$

Define f(n,k)

```
fnk = 1/nn (2^(1-2 nn) k (1 - 2 k + 2 nn))/(-1 + 2 k) (2 k)!/(k! k!) (2 nn - 2 k)!/((nn - k)! (nn - k)!)
```

$$\frac{2^{1-2\,nn}\,k\,(1-2\,k+2\,nn)\,(2\,k)!\,(-2\,k+2\,nn)!}{(-1+2\,k)\,nn\,(k!)^2\,((-k+nn)!)^2}$$

```
res = Zb[fnk, {k, 1, nn}, nn, 1]
```

```
If `-1 + nn' is a natural number, then:
{SUM[nn] - SUM[1 + nn] == 0}
```

this tells us that the summation is independent of nn. We check for a value of nn=1 and do the summation (which is just k=1)

```
fnk /. {nn → 1, k → 1}
```

```
1
```

now get the proof certificate

```
proofCert = show[R]
```

$$\frac{(-1+k)\,(3-2\,k+2\,nn)}{2\,(1+nn)\,(1-k+nn)}$$

get the g function, such that f(n+1,k)-f(n,k) == g(n,k+1)-g(n,k)

```
gnk = proofCert × fnk
```

$$\frac{2^{-2\,nn}\,(-1+k)\,k\,(1-2\,k+2\,nn)\,(3-2\,k+2\,nn)\,(2\,k)!\,(-2\,k+2\,nn)!}{(-1+2\,k)\,nn\,(1+nn)\,(1-k+nn)\,(k!)^2\,((-k+nn)!)^2}$$

verify that f(n+1,k)-f(n,k) == g(n,k+1)-g(n,k)

```
fnk - (fnk /. nn → nn + 1) == (gnk /. k → k + 1) - gnk // FullSimplify
```

```
True
```

show that the gnk is zero for a finite top and bottom

```
gnk /. k → 0
```

```
0
```

```
gnk /. k → nn + 2
```

```
0
```

now show proof

```
Prove[]
```

Theorem

Let

$$F(k,\ \mathrm{nn}) = \frac{k\, 2^{1-2\,\mathrm{nn}}\,(2\,k)!\,(-2\,k+2\,\mathrm{nn}+1)\,(2\,\mathrm{nn}-2\,k)!}{(2\,k-1)\,\mathrm{nn}\,(k!)^2\,((\mathrm{nn}-k)!)^2}$$

and

$$\mathrm{SUM}(\mathrm{nn}) = \sum_{k=1}^{\mathrm{nn}} F(k,\ \mathrm{nn}).$$

If `-1 + nn' is a natural number, then:

$$\mathrm{SUM}(\mathrm{nn}) - \mathrm{SUM}(1+\mathrm{nn}) = 0.$$

Proof

Let Δ_k(".") denote the forward difference operator in *k* and define

$$R(k,\ \mathrm{nn}) = \frac{(k-1)\,(-2\,k+2\,\mathrm{nn}+3)}{2\,(\mathrm{nn}+1)\,(-k+\mathrm{nn}+1)}.$$

Then the Theorem follows from summing the equation

$$F(k,\ \mathrm{nn}) - F(k,\ 1+\mathrm{nn}) = \Delta_k(F(k,\ \mathrm{nn})\,R(k,\ \mathrm{nn}))$$

over *k* from 1 to nn + 1.

This equation is routinely verifyable by dividing both sides by *F* and checking the resulting rational equation.

Prove that the row sums terms are correct

We are proving that $H\,(H^{-1}\,e) = e$

```
fnk = 1/(2 (i - k) + 1) * (2^(1-2 nn) k (1 - 2 k + 2 nn))/(-1 + 2 k) * (2 k)!/(k! k!) * (2 nn - 2 k)!/((nn - k)! (nn - k)!)
```

$$\frac{2^{1-2\,nn}\,k\,(1-2k+2\,nn)\,(2k)!\,(-2k+2\,nn)!}{(1+2\,(i-k))\,(-1+2k)\,(k!)^2\,((-k+nn)!)^2}$$

```
res = Zb[(fnk), {k, 1, nn}, nn, 2] // First
If `-1 + nn' is a natural number, then:
```

$$2\,(-1+i-nn)\,\mathrm{SUM}[nn]+(5-4i+4\,nn)\,\mathrm{SUM}[1+nn]+(-3+2i-2\,nn)\,\mathrm{SUM}[2+nn]==0$$

verify for base case of nn=1 (and therefore i=1 and k=1)

```
base = Sum[fnk /. {nn → 1, i → 1}, {k, 1, 1}]
```

1

verify for nn = 2, for both i=1 and i=2

```
basep1 = Sum[fnk /. {nn → 2, i → 1}, {k, 1, 2}]
```

1

```
basep1 = Sum[fnk /. {nn → 2, i → 2}, {k, 1, 2}]
```

1

then show for all values, the induction step

```
Solve[2 (-1 + i - nn) + (5 - 4 i + 4 nn) + (-3 + 2 i - 2 nn) s2 == 0, s2]
```

{{s2 → 1}}

now get the proof certificate

```
rnk = show[R]
```

$$\frac{(-1+k)\,(-1-2i+2k)\,(3-2k+2\,nn)}{4\,(1+nn)\,(1-k+nn)\,(2-k+nn)}$$

The theorem

```
Prove[]
```

and

$$\mathrm{SUM}(nn)=\sum_{k=1}^{nn}F(k,\,nn).$$

If `-1 + nn' is a natural number, then:

$$2\,(i-nn-1)\,\mathrm{SUM}(nn)+(-4\,i+4\,nn+5)\,\mathrm{SUM}(1+nn)+(2\,i-2\,nn-3)\,\mathrm{SUM}(2+nn)=0.$$

The proof

Let Δ_k(".") denote the forward difference operator in k and define

$$R(k,\ \text{nn}) = \frac{(k-1)\,(-2\,i+2\,k-1)\,(-2\,k+2\,\text{nn}+3)}{4\,(\text{nn}+1)\,(-k+\text{nn}+1)\,(-k+\text{nn}+2)}.$$

Then the Theorem follows from summing the equation

$$2\,(i-\text{nn}-1)\,F(k,\ \text{nn}) + (-4\,i+4\,\text{nn}+5)\,F(k,\ 1+\text{nn}) + (2\,i-2\,\text{nn}-3)\,F(k,\ 2+\text{nn}) = \Delta_k(F(k,\ \text{nn})\,R(k,\ \text{nn}))$$

over k from 1 to nn + 2.

This equation is routinely verifyable by dividing both sides by F and checking the resulting rational equation.

The verification

```
gnk = rnk fnk
```

$$\frac{2^{-1-2\,\text{nn}}\,(-1+k)\,k\,(-1-2\,i+2\,k)\,(1-2\,k+2\,\text{nn})\,(3-2\,k+2\,\text{nn})\,(2\,k)!\,(-2\,k+2\,\text{nn})!}{(1+2\,(i-k))\,(-1+2\,k)\,(1+\text{nn})\,(1-k+\text{nn})\,(2-k+\text{nn})\,(k!)^2\,((-k+\text{nn})!)^2}$$

```
gnkplus1 = gnk /. k → k + 1
```

$$\left(2^{-1-2\,\text{nn}}\,k\,(1+k)\,(-1-2\,i+2\,(1+k))\,(1-2\,(1+k)+2\,\text{nn})\,(3-2\,(1+k)+2\,\text{nn})\,(2\,(1+k))!\,(-2\,(1+k)+2\,\text{nn})!\right) \Big/ \left((1+2\,(-1+i-k))\,(-1+2\,(1+k))\,(1+\text{nn})\,(-k+\text{nn})\,(1-k+\text{nn})\,((1+k)!)^2\,((-1-k+\text{nn})!)^2\right)$$

```
fnknplus1 = fnk /. nn → nn + 1
```

$$\frac{2^{1-2\,(1+\text{nn})}\,k\,(1-2\,k+2\,(1+\text{nn}))\,(2\,k)!\,(-2\,k+2\,(1+\text{nn}))!}{(1+2\,(i-k))\,(-1+2\,k)\,(k!)^2\,((1-k+\text{nn})!)^2}$$

```
fnknplus2 = fnk /. nn → nn + 2
```

$$\frac{2^{1-2\,(2+\text{nn})}\,k\,(1-2\,k+2\,(2+\text{nn}))\,(2\,k)!\,(-2\,k+2\,(2+\text{nn}))!}{(1+2\,(i-k))\,(-1+2\,k)\,(k!)^2\,((2-k+\text{nn})!)^2}$$

Confirm the recursion formula

```
Assuming[{i, nn} ∈ Integers,
 (2 (i - nn - 1) fnk/fnk + (-4 i + 4 nn + 5) fnknplus1/fnk + (2 i - 2 nn - 3) fnknplus2/fnk == gnkplus1/fnk - gnk/fnk) //
  FullSimplify]
```

```
True
```

Confirm the telescoping terms are zero

```
gnk /. k → 0
```

```
0
```

```
gnk /. k → nn + 3
```

```
0
```